

The REBOL Documentation Project

-- FR - Documentation REBOL - HOWTOs --

HOWTOs

Comment faire pour ...

REBOLtoF

Première publication : 11 mai 2005, et mis en
ligne le mercredi 29 mars 2006

Résumé :

Vous trouverez dans cet "HowTo" quelques trucs et astuces utilisés par les programmeurs REBOL expérimentés.

Ceux-ci proviennent d'expériences personnelles, mais également des diverses conversations ayant pris place sur les forums et mailing-list de REBOL.

Vous trouverez dans cet "HowTo" quelques trucs et astuces utilisés par les programmeurs REBOL expérimentés.

Ceux-ci proviennent d'expériences personnelles, mais également des diverses conversations ayant pris place sur les forums et mailing-list de REBOL.

Vous connaissez aussi un truc utile ? N'hésitez pas à nous en faire part : le formulaire de la rubrique [Contribution / Aide](#) est à votre disposition !

Table des matières

- [1 ... créer facilement un identifiant unique ?](#)
- [2 ... mettre à jour les ascenseurs du widget "text-list" ?](#)
- [3 ... rendre VID moins gourmand en mémoire ?](#)
- [4 ... vider la file d'attente \(queue\) des 'events ?](#)
- [5 ... parcourir un répertoire et ses sous-répertoires en appliquant un traitement à ses fichiers ?](#)
- [6 ...créer une variable prenant des valeurs alternatives ?](#)
- [7 ...faire figurer des espaces dans les noms de fichiers ?](#)
- [8 ...faire figurer une devise locale dans de datatype money! ?](#)
- [9 ...connaître l'adresse IP de sa machine ?](#)
- [10 ...sous Linux, évaluer correctement un script REBOL rendu exécutable ?](#)
- [11 ...connaître le répertoire courant ?](#)
- [12 ...convertir une chaîne de caractères en un nombre ?](#)
- [13 ...rendre un script exécutable sous Unix et Linux ?](#)
- [14 ...mettre en surbrillance une ligne dans un style text-list ?](#)
- [15 ...désactiver la connexion automatique ?](#)
- [16 ...afficher une boîte d'alerte ?](#)
- [17 ...visualiser le code source de view ?](#)
- [18 ...centrer une fenêtre sur l'écran ?](#)
- [19 ...obtenir la taille du bureau ?](#)
- [20 ...cacher le texte d'un champ de saisie ?](#)
- [21 ...traduire les jours et les mois dans le calendrier de Rebol/View ?](#)
- [22 ...connaître les dimensions d'un texte à l'écran ?](#)
- [23 ...sauvegarder un layout dans un fichier graphique et l'afficher ensuite dans un navigateur ?](#)
- [24 ...trouver le numéro de licence de REBOL/Command ?](#)
- [25 ... enlever sélectivement des éléments d'un block! ?](#)
- [26 ... déclarer au sein d'un objet des membres *réellement* privés ?](#)
- [27 ... transformer une date dans un format semblable à 02/05/2005 ?](#)
- [28 ... charger les données d'un fichier externes \(.ini ou . cfg par exemple\) ?](#)
- [29 ... extraire une chaîne de caractères d'une autre chaîne ?](#)
- [30 ... trier les données d'un block! sur une longueur fixe ?](#)
- [31 ... supprimer l'action du click droit de la souris sur un bouton ?](#)
- [32 ... créer dynamiquement le nom d'une variable ?](#)
- [33 ... créer dynamiquement des tag! ?](#)

- [34 ... forcer la libération de mémoire ?](#)
- [35 ... insérer un certain nombre de fois un caractère dans une chaîne de caractères ?](#)
- [36 ... tester l'emploi de *refinements* mutuellement exclusifs ?](#)
- [37 ... créer dynamiquement un *refinement* d'une fonction à partir d'une variable ?](#)

1 ... créer facilement un identifiant unique ?

par REBOLtoF - 29 mars 2006

Une technique que j'emploie régulièrement afin de créer des identifiants uniques (par exemple pour un nom de fichier), est la suivante :

```
>> timestamp: form now/precise
== "24-Mar-2006/9:41:07.154+1:00"
>> foreach elem ["-" "/" ":" "." "+"] [replace/all timestamp elem ""]
== "24Mar200694107154100"
```

ou, en employant la nouvelle fonction 'remove-each :

```
>> timestamp: form now/precise
== "24-Mar-2006/9:49:20.664+1:00"
>> remove-each elem timestamp [find ["-" "/" ":" "." "+"] elem]
== "24Mar200694920664100"
```

Eventuellement, afin d'éviter les collisions dues à des horloges non-synchronisées, un nombre random peut être rajouté à l'id générée...

2 ... mettre à jour les ascenseurs du widget "text-list" ?

par REBOLtoF - 23 mars 2006

Dans /View 1.3, la solution est très simple :

```
ma-text-list/update
show ma-text-list
```

Pour les versions précédentes, il est nécessaire de coder soi-même la fonctionnalité. Par exemple

```
either 0 = length? ma-text-list/data [
    ma-text-list/sld/redrag 1
] [
    ma-text-list/sld/redrag ma-text-list/lc / length? ma-text-list/data
]
show ma-text-list
```

Dans les deux cas, l'effet sera que la taille du curseur de l'ascenseur s'adaptera au nombre d'éléments présents dans la text-list.

3 ... rendre VID moins gourmand en mémoire ?

par REBOLtoF - 26 janvier 06

Afin d'illustrer le problème de la gourmandise de VID en terme de RAM, nous avons mesuré récemment que la génération de quelque 200 fenêtres contenant des images prenais 60 MB de RAM, *et ne les rendait pas*. Sur certains hardwares anciens, cela se termine par souvent un crash système !

Et oui, VID est verbeux. Et cela est dû au modèle utilisé afin de créer les 'face : toutes se basent sur le template de 'vid-face :

```
system/view/VID/vid-face
```

Or, toutes les propriétés (ou 'facets) de la 'face ne sont pas utilisées dans tous les cas... Nous avons par exemple parfois besoin d'un simple 'text, auquel cas les 'feel, 'edge et autre 'para ne sont pas utiles...

Nous avons alors deux solutions :

1. lors de la définition de la 'face, mettre explicitement les 'facets inutiles à 'none :

```
view layout [  
  text "hello" with [feel: none para: none edge: none]  
]
```

2. quand nous n'avons plus besoin de notre 'face (par exemple, lors d'un changement de fenêtre), explicitement mettre TOUTES les 'facets à none, de manière dynamique. Si 'workzone constitue le 'pane de notre écran principal, cela donne :

```
foreach elem workzone/pane/pane [  
  foreach facet remove first elem [  
    set in elem facet none  
  ]  
]
```

En appliquant cette dernière méthode, nous avons fait chuter la consommation mémoire de 60 MB à 1.8 MB ...

4 ... vider la file d'attente (queue) des 'events ?

par REBOLtoF - 26 janvier 06

Dans certains cas spécifiques, il peut être nécessaire de vider la file d'attente des 'events, par exemple afin de s'assurer que plus aucun 'event ne reste en attente de traitement avant la génération d'une nouvelle fenêtre avec /View...

Cette file d'attente est en fait un 'port en REBOL, mais il n'est pas permis d'utiliser le 'clear pour la nettoyer...

Le truc consiste à la parcourir, event par event, en utilisant un 'pick jusqu'à ce qu'on rencontre un 'none (qui signifie la fin de la file) :

```
until [none? pick system/view/event-port 1]
```

Cependant, attention : tous les 'events en attente (comme presser une touche ou un click de souris) sont alors perdus !

5 ... parcourir un répertoire et ses sous-répertoires en appliquant un traitement à ses fichiers ?

par REBOLtoF - 06 Jan 06

En utilisant une fonction récurrente évidemment :-)

Par exemple, cette fonction 'scan :

```
scan: func [dir][
    ;--- parcourt tout les éléments (fichier ou répertoires)
    ;- trouvés dans le répertoire passé en argument
    foreach elem read dir [
        ;--- si l'élément est un répertoire...
        either #"/" = last form elem [
            ;--- ... alors appel récurrent à la fonction
            scan join dir elem
        ][
            ;--- ... sinon appliquer le traitement voulu
            ;- à l'élément (fichier) trouvé
            attempt [do action elem]
        ]
    ]
]
```

Remarquez que, à la place du test *either #"/" = last form elem* nous pourrions utiliser *either dir ? elem*. Cependant, pour des raisons que j'ignore encore, cette fonction ne renvoie pas toujours les résultats attendus.

Ici, le traitement à appliquer aux fichiers trouvés a été isolé, afin d'offrir plus de souplesse au script.

Ce traitement peut par exemple être "compter le nombre de fichiers PDF trouvés" :

```
count: 0
action: func [elem] [
    if find elem %.pdf [
        count: count + 1
    ]
]
```

Il nous suffit alors de repérer le répertoire de base, et de lancer le scan :

```
scan %mon-repertoire/
```

REBOL, keep IT simple :-)

6 ...créer une variable prenant des valeurs alternatives ?

par REBOLtoF - 23 decembre 05

Imaginons que nous ayons une variable devant prendre deux valeurs : celles des couleurs 'white et 'black. Comment faire pour que cette variable prenne la valeur 'black si elle contenait 'white, et la valeur 'white si elle contenait 'black ?

La solution la plus simple est basée sur l'utilisation de la fonction 'pick :

```
>> ma-var: white ==> il faut lui donner une valeur initiale
== 255.255.255
>> ma-var: pick [white black] ma-var = 'black
== black
>> ma-var: pick [white black] ma-var = 'black
== white
>> ma-var: pick [white black] ma-var = 'black
== black
```

Le code "ma-var = 'black" renvoie un logic!, qui est pris comme argument pour la fonction 'pick !

7 ...faire figurer des espaces dans les noms de fichiers ?

par REBOLtoF - 17 octobre 05

Deux solutions existent :

1. Utilisation du code hexadécimal :

```
fichier: %/mes%20documents/monfichier.r
```

2. Mise entre guillemets :

```
fichier: %"/mes documents/monfichier.r"
```

8 ...faire figurer une devise locale dans de datatype money! ?

par REBOLtoF - 17 octobre 05

Tout simplement en le faisant précéder le money! :

```
>> argent: EUR$26.00
== EUR$26.00
>> type? argent
== money!
>> probe argent
EUR$26.00
== EUR$26.00
>> first argent
== "EUR"
```

```
>> second argent  
== 26.0
```

9 ...connaître l'adresse IP de sa machine ?

par Olivier Martin - 31 juillet 05

```
read make url! join "dns://" (read dns://)
```

10 ...sous Linux, évaluer correctement un script REBOL rendu exécutable ?

par Olivier Martin - 31 juillet 05

Ce problème survient généralement lorsqu'on tente d'utiliser sur une machine Linux un script écrit sur une machine Windows. A l'aide de l'éditeur VI, vous pouvez savoir si le fichier est au format DOS. Si c'est le cas, vous devez utiliser un convertisseur tel que dos2unix pour éliminer les caractères de contrôle qui perturbent l'évaluateur.

11 ...connaître le répertoire courant ?

par Olivier Martin - 31 juillet 05

La fonction what-dir contient le répertoire actif

12 ...convertir une chaîne de caractères en un nombre ?

par Olivier Martin - 31 juillet 05

Vous devez évaluer avec le mot `do` la chaîne de caractères contenant le nombre. Vous pouvez ensuite l'affecter à une variable :

```
v: do "10"
```

13 ...rendre un script exécutable sous Unix et Linux ?

par Olivier Martin - 31 juillet 05

La première ligne du fichier doit préciser le chemin d'accès à l'évaluateur utilisé et les options d'exécution (ex : `#!/usr/bin rebol -cs`). L'instruction `chmod +x` permet ensuite de transformer le script en un fichier exécutable.

14 ...mettre en surbrillance une ligne dans un style text-list ?

par Olivier Martin - 31 juillet 05

`pindex: 0`

```
affiche-liste: make function! [] [
    tclear liste/picked
    tchaine: copy []
    tchaine2: pick liste/data pindex
    tinsert chaine pick liste/data pindex
    tinsert liste/picked chaine
    tshow liste
]

view layout [
    tliste: text-list data [ "ligne1" "ligne2" "ligne3" ] [
        ttpindex: face/cnt
        t
    ]
    treturn
    tbutton "Haut" [
        ttif pindex > 1 [ pindex: pindex - 1 ]
        ttaffiche-liste
        t
    ]
    tbutton "Bas" [
        ttif pindex
            ttaffiche-liste
        t
    ]
]
```

15 ...désactiver la connexion automatique ?

par Olivier Martin - 31 juillet 05

Pour empêcher Rebol/View de se connecter à un serveur distant dès son démarrage, vous devez modifier le paramètre auto-connect du fichier de configuration prefs.r en lui assignant la valeur false.

16 ...afficher une boîte d'alerte ?

par Olivier Martin - 31 juillet 05

`alert "ceci est mon message"`

17 ...visualiser le code source de view ?

par Olivier Martin - 31 juillet 05

Il suffit d'afficher l'objet "system/view" dans un fichier texte :

```
save %/c/view.txt system/view
```

18 ...centrer une fenêtre sur l'écran ?

par Olivier Martin - 31 juillet 05

```
view center-face layout [text "Salut" ]
```

19 ...obtenir la taille du bureau ?

par Olivier Martin - 31 juillet 05

```
system/view/screen-face/size
```

20 ...cacher le texte d'un champ de saisie ?

par Olivier Martin - 31 juillet 05

```
view layout [text "Mot de passe" field hide]
```

21 ...traduire les jours et les mois dans le calendrier de Rebol/View ?

par Olivier Martin - 31 juillet 05

Vous devez modifier les propriétés system/locale/days et system/locale/months.

REBOL [Title : "la date en français" Author : "FJouen" Version 1.0]

```
system/locale/months : ["janvier" "Février" "Mars" "Avril" "Mai" "Juin" "Juillet" "Août" "Septembre"  
"Octobre" "Novembre" "Décembre"]
```

```
system/locale/days: ["lundi" "mardi" "mercredi" "jeudi" "vendredi" "samedi" "dimanche"]
```

```
request-date
```

22 ...connaître les dimensions d'un texte à l'écran ?

par Olivier Martin - 31 juillet 05

Le mot `size-text`, suivi du nom du style `text`, retourne les dimensions de celui-ci.

23 ...sauvegarder un layout dans un fichier graphique et l'afficher ensuite dans un navigateur ?

```
save/png %toto.png to-image nomdulayout_a_capturer
```

```
browse %toto.png
```

24 ...trouver le numéro de licence de REBOL/Command ?

par Olivier Martin - 31 juillet 05

Le numéro de licence est le fichier `licence.key`. Il doit se trouver dans le répertoire courant de votre application. Pour les hébergeurs de services, ceci permet d'avoir plusieurs licences d'utilisation sur une seule machine.

25 ... enlever sélectivement des éléments d'un block! ?

par REBOLtof - 23 mai 2005

REBOL nous offre pour cela une fonction spécifique, qui gagne à être connue : `'remove-each`. Celle-ci prend comme argument un `word!`, le `block!` lui-même, et un `body (block!)` dans lequel nous pouvons faire figurer la condition qui, vérifiée, devra autoriser la suppression de l'élément. Supposons que nous voulions supprimer toutes les strings de ce bloc :

```
>> mon-block: [a "a" b "b" c "c"]
== [a "a" b "b" c "c"]
>> remove-each element mon-block [string? element]
== [a b c]
```

Vous pouvez évidemment faire figurer n'importe quelle condition dans le dernier bloc...

26 ... déclarer au sein d'un objet des membres *réellement* privés ?

par REBOLtof - 21 septembre 2001

Le modèle objet de REBOL n'est pas aussi complet comme celui de Java ou de Python. Cependant, en jouant sur certaines fonctions particulières, nous avons la possibilité d'empêcher nos variables d'être accédées de l'extérieur d'un objet. Nous allons utiliser pour cela les fonctions `'use` et `'set` :

```
use [var-priv-1 var-priv-2 func-priv][
  mon-objet: make object! [
```

```

    ;--> variables privées
    set 'var-priv-1 1
    set 'var-priv-2 2

    ;--> variable publique
    var-pub: 5

    ;--> fonction privée
    set 'func-priv does [var-priv-1 + var-priv-2]

    ;--> fonctions publiques
    func-pub-1: does [10 * var-pub]
    func-pub-2: does [func-priv]
  ]
]

```

Ainsi, seuls 'var-pub, 'func-pub-1 et 'func-pub-2 sont visibles de l'extérieur :

```

>> mon-objet/var-pub
== 5
>> mon-objet/func-pub-1
== 50
>> mon-objet/func-pub-2
== 3

```

Tandis que les membres privés restent protégés :

```

>> mon-objet/var-priv-1
** Script Error: Invalid path value: var-priv-1
** Near: mon-objet/var-priv-1
>> mon-objet/var-priv-2
** Script Error: Invalid path value: var-priv-2
** Near: mon-objet/var-priv-2
>> mon-objet/func-priv
** Script Error: Invalid path value: func-priv
** Near: mon-objet/func-priv

```

27 ... transformer une date dans un format semblable à 02/05/2005 ?

par johnatemp - 01 février 05

Cette petite fonction le permet :

```

formater-date: function [dat [date!] /int] [d data j] [
  data: [
    "Jan" "01" "Feb" "02" "Mar" "03" "Apr" "04"
    "May" "05" "Jun" "06" "Jul" "07" "Aug" "08"
    "Sep" "09" "Oct" "10" "Nov" "11" "Dec" "12"
  ] ; définition du tableau de correspondance mois/numéro
  d: parse to-string dat "-" ; découpe de la date en 3 éléments
  either (length? first d)      [j: join "0" first d] ; ajouter un 0
  [[j: first d]
  return join j ["/" select data second d "/" third d] ;formater la date, en utilisant la

```

```
table de correspondance
```

```
]
```

```
Exemple :>> formater-date 1/1/1
```

```
== "01/01/2001"
```

```
>> formater-date 12-4-98
```

```
== "12/04/1998"
```

```
>> formater-date 2-dec-1998
```

```
== "02/12/1998"
```

28 ... charger les données d'un fichier externes (.ini ou .cfg par exemple) ?

Une approche simple consiste à évaluer ('do) un block chargé par la fonction 'load. Le 'load retournant toujours un block ! et celui-ci protégeant son contenu de l'évaluation, il est en effet nécessaire de forcer cette évaluation par l'emploi de 'do :

```
do load %prefs.ini
```

Une forme plus sophistiquée (*par ViE - 14 mai 05*) consiste à charger les données du fichier directement dans un objet (ou contexte) afin de les protéger d'une redéfinition :

```
prefs: context load %prefs.ini
```

Cette méthode présente un avantage supplémentaire : fournir des paramètres par défaut (en cas d'absence, ou de version incomplète du fichier) en initialisant l'objet :

```
; les paramètres par défaut
```

```
prefs: context [
```

```
  info1: ["ville1"]
```

```
  info2: ["ville1" [temps1: "2h" temps2: "2h30"]]
```

```
  détail: "Pas de détail" ; valeur par défaut
```

```
]
```

```
; si le fichier existe, on charge les paramètres qui y sont définis
```

```
if exists? %prefs.ini [prefs: make prefs load %prefs.ini]
```

29 ... extraire une chaîne de caractères d'une autre chaîne ?

Plusieurs solutions existent :

```
substring: func [chaîne pos-début longueur-substring] [
```

```
  copy/part at chaîne pos-début longueur-substring
```

```
]par darkblue
```

```
substring: func [chaîne pos-début pos-fin] [
```

```
  copy/part at chaîne pos-début pos-fin - pos-début + 1
```

```
]par darkblue
```

```
substring: func [chaîne pos-début pos-fin] [
```

```
  copy/part at chaîne pos-début add 1 subtract pos-fin pos-début
```

```
]par Bouba - identique au précédent, sauf qu'il n'utilise pas les opérateurs in-lines
```

```
Exemple :>> ma-chaîne: "ceci est une phrase sans aucun sens mais longue"
```

```
== "ceci est une phrase sans aucun sens mais longue"
>> substring: func [chaine pos-début pos-fin] [
  [      copy/part at chaine pos-début add 1 subtract pos-fin pos-début
    [      ]
  ]
>> substring str 5 10
== " est u"
```

30 ... trier les données d'un block! sur une longueur fixe ?

par REBOLtoF

On peut utiliser le 'sort/skip

Exemple :>> b: [a 1 c 5 r 8 b 2 a 9 g 4]

```
== [a 1 c 5 r 8 b 2 a 9 g 4]
```

```
>> b1: copy b
```

```
== [a 1 c 5 r 8 b 2 a 9 g 4]
```

```
>> sort b1
```

```
== [a a b c g r 1 2 4 5 8 9]
```

```
>> b1
```

```
== [a a b c g r 1 2 4 5 8 9]
```

```
>> b1: copy b
```

```
== [a 1 c 5 r 8 b 2 a 9 g 4]
```

```
>> sort/skip b1 2
```

```
== [a 1 a 9 b 2 c 5 g 4 r 8]
```

31 ... supprimer l'action du click droit de la souris sur un bouton ?

par REBOLtoF

Le bouton (btn ou button ou toggle) accepte les click droits et gauche afin d'être activé.

Le code peut être du type :

```
view layout [
  btn "text" ["action click gauche"] ["action click droit"]
]
```

L'état du bouton (enfoncé ou non) est reflété par la propriété :

`face/state`

qui est un logic ! .

Empêcher l'action du bouton droit de la souris est réalisé par :

```
view layout [
  btn "text" ["action click gauche"] [face/state: not face/state]
]
```

32 ... créer dynamiquement le nom d'une variable ?

par *Pedro*

En utilisant une combinaison des fonctions 'set et 'to-word :

```
>> nom-1: 'mon-nombre
== mon-nombre
>> nom-2: 'magique
== magique
>> set to-word rejoin [nom-1 "-" nom-2] 100
== 100
>> ? mon-nombre-magique
MON-NOMBRE-MAGIQUE is an integer of value: 100
```

33 ... créer dynamiquement des tag! ?

par *REBOLtoF*

Nous utilisons pour cela les fonctions 'join ou 'rejoin. Mais ces deux fonctions utilisent leur premier argument comme datatype par défaut :

```
>> join "example" 123
== "example123"

>> rejoin [http://mon-server/ "index.html"]
== http://mon-server/index.html
```

Mais dans notre cas, ce comportement ne satisfait pas :

```
>> join ["Page"]
== >
```

```
>> rejoin ["Page"]
== >
```

Une solution possible est l'ajout d'un string vide :

```
>> join "" ["Page"]
== {Page}
```

```
>> rejoin ["" "Page"]
== {Page}
```

Ou :

```
<code>>> to-string ["Page"]
== {Page}
```

34 ... forcer la libération de mémoire ?

par *DocKimbel*

Il nous faut utiliser une combinaison des fonctions 'make et 'recycle :

```
free-mem: func ['word] [set :word make none! recycle]
```

Exemple :>> `system/stats`

```
== 3778424
>> s: make string! 50'000'000
== ""
>> system/stats
== 53780344
>> free-mem s
>> system/stats
== 3778424
```

35 ... insérer un certain nombre de fois un caractère dans une chaîne de caractères ?

par Pablo

En utilisant la fonction `insert` qui dispose du *refinement* `/dup` (*duplicate*). Cette fonctionnalité est utile pour, par exemple, insérer un certain nombre de blancs en tête de chaîne :

```
>> ma-chaîne: "unmot"
== "unmot"
>> insert/dup ma-chaîne " " 10
== "unmot"
>> head ma-chaîne
== "          unmot"
```

La même technique est utilisée afin d'ajouter en fin de chaîne :

```
>> ma-chaîne: "unmot"
== "unmot"
>> insert/dup tail ma-chaîne "t" 5
== "">> head ma-chaîne
== "unmotttttt"
```

Remarquez que cette technique fonctionne avec tous les datatypes *series*!, donc également les *block*!

36 ... tester l'emploi de *refinements* mutuellement exclusifs ?

par Yos

Nous pouvons jouer sur les propriétés de la fonction `'switch` :

```
ma-fonction: func [
    /ref1 /ref2 /ref3
][
    print switch/default true reduce [
        ref1 ["ref1"]
        ref2 ["ref2"]
        ref3 ["ref3"]
    ]
```

```
    ]["pas de raffinement"]  
  ]
```

Ce qui nous donne :

```
>> ma-fonction  
pas de raffinement  
>> ma-fonction/ref1  
ref1  
>> ma-fonction/ref2  
ref2  
>> ma-fonction/ref3  
ref3
```

37 ... créer dynamiquement un *refinement* d'une fonction à partir d'une variable ?

par Marco

Comme en REBOL il est possible de générer du code à la volée, la solution est relativement simple :

```
>> ma-fonction: func [arg /ref1][if ref1 [print "ref1"] 1 + arg]  
>> ma-fonction/ref1 5  
ref1  
== 6  
>> mon-refinement: 'ref1  
== ref1  
>> do compose [(join to-path 'ma-fonction mon-refinement) 10]  
ref1  
== 11
```